Rules for integrands of the form $(d\,x)^m\,(a+b\,\text{ArcCosh}[c\,x])^n$

1. $\int (d\,x)^m\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx$ when $n\in\mathbb{Z}^+$

1: $\int \frac{(a+b\,\text{ArcCosh}[c\,x])^n}{x}\,dx$ when $n\in\mathbb{Z}^+$

Derivation: Integration by substitution

Basis: $\frac{1}{x} = \frac{1}{b}\,\text{Subst}\left[\text{Tanh}\left[-\frac{a}{b}+\frac{x}{b}\right],\,x,\,a+b\,\text{ArcCosh}[c\,x]\right]\,\partial_x\left(a+b\,\text{ArcCosh}[c\,x]\right)$

Note: If $n\in\mathbb{Z}^+$, then $x^n\,\text{Tanh}\left[-\frac{a}{b}+\frac{x}{b}\right]$ is integrable in closed-form.

Rule: If $n\in\mathbb{Z}^+$, then

$$\int \frac{(a+b\,\text{ArcCosh}[c\,x])^n}{x}\,dx \longrightarrow \frac{1}{b}\,\text{Subst}\left[\int x^n\,\text{Tanh}\left[-\frac{a}{b}+\frac{x}{b}\right]dx,\,x,\,a+b\,\text{ArcCosh}[c\,x]\right]$$

Program code:

```
Int[(a_.+b_.*ArcCosh[c_.*x_])^n_./x_,x_Symbol] :=
  1/b*Subst[Int[x^n*Tanh[-a/b+x/b],x],x,a+b*ArcCosh[c*x]] /;
FreeQ[{a,b,c},x] && IGtQ[n,0]
```

2: $\int (d\,x)^m\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx$ when $n\in\mathbb{Z}^+\land m\neq -1$

Derivation: Integration by parts

Basis: $\partial_x\,(a+b\,\text{ArcCosh}[c\,x])^n == \frac{b\,c\,n\,(a+b\,\text{ArcCosh}[c\,x])^{n-1}}{\sqrt{1+c\,x}\,\sqrt{-1+c\,x}}$

Rule: If $n\in\mathbb{Z}^+\land m\neq -1$, then

$$\int (d\,x)^m\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx\longrightarrow\frac{(d\,x)^{m+1}\,(a+b\,\text{ArcCosh}[c\,x])^n}{d\,(m+1)}-\frac{b\,c\,n}{d\,(m+1)}\int\frac{(d\,x)^{m+1}\,(a+b\,\text{ArcCosh}[c\,x])^{n-1}}{\sqrt{1+c\,x}\,\sqrt{-1+c\,x}}\,dx$$

Program code:

```
Int[(d_.*x_)^m_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (d*x)^(m+1)*(a+b*ArcCosh[c*x])^n/(d*(m+1)) -
  b*c*n/(d*(m+1))*Int[(d*x)^(m+1)*(a+b*ArcCosh[c*x])^(n-1)/(Sqrt[1+c*x]*Sqrt[-1+c*x]),x] /;
FreeQ[{a,b,c,d,m},x] && IGtQ[n,0] && NeQ[m,-1]
```

2. $\int x^m\,(a+b\,\mathrm{ArcCosh}[c\,x])^n\,dx$ when $m\in\mathbb{Z}^+$

1: $\int x^m\,(a+b\,\mathrm{ArcCosh}[c\,x])^n\,dx$ when $m\in\mathbb{Z}^+\land n>0$

Derivation: Integration by parts

Basis: $\partial_x\,(a+b\,\mathrm{ArcCosh}[c\,x])^n = \frac{b\,c\,n\,(a+b\,\mathrm{ArcCosh}[c\,x])^{n-1}}{\sqrt{1+c\,x}\,\sqrt{-1+c\,x}}$

Rule: If $n\in\mathbb{Z}^+\land m\neq-1$, then

$$\int x^m\,(a+b\,\mathrm{ArcCosh}[c\,x])^n\,dx\;\longrightarrow\;\frac{x^{m+1}\,(a+b\,\mathrm{ArcCosh}[c\,x])^n}{m+1}-\frac{b\,c\,n}{m+1}\int\frac{x^{m+1}\,(a+b\,\mathrm{ArcCosh}[c\,x])^{n-1}}{\sqrt{1+c\,x}\,\sqrt{-1+c\,x}}\,dx$$

Program code:

```
Int[x_^m_.*(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  x^(m+1)*(a+b*ArcCosh[c*x])^n/(m+1) -
  b*c*n/(m+1)*Int[x^(m+1)*(a+b*ArcCosh[c*x])^(n-1)/(Sqrt[1+c*x]*Sqrt[-1+c*x]),x] /;
FreeQ[{a,b,c},x] && IGtQ[m,0] && GtQ[n,0]
```

2. $\int x^m\,(a+b\,\mathrm{ArcCosh}[c\,x])^n\,dx$ when $m\in\mathbb{Z}^+\land n<-1$

1: $\int x^m\,(a+b\,\mathrm{ArcCosh}[c\,x])^n\,dx$ when $m\in\mathbb{Z}^+\land -2\le n<-1$

Derivation: Integration by parts and integration by substitution

Basis: $\frac{(a+b\,\mathrm{ArcCosh}[c\,x])^n}{\sqrt{1+c\,x}\,\sqrt{-1+c\,x}} = \partial_x\,\frac{(a+b\,\mathrm{ArcCosh}[c\,x])^{n+1}}{b\,c\,(n+1)}$

Basis: $\partial_x\left(x^m\,\sqrt{1+c\,x}\,\sqrt{-1+c\,x}\right) = -\frac{x^{m-1}\,\left(m-(m+1)\,c^2\,x^2\right)}{\sqrt{1+c\,x}\,\sqrt{-1+c\,x}}$

Basis: $\frac{F[x]}{\sqrt{1+c\,x}\,\sqrt{-1+c\,x}} = \frac{1}{b\,c}\,\mathrm{Subst}\left[F\left[\frac{\mathrm{Cosh}\left[-\frac{a}{b}+\frac{x}{b}\right]}{c}\right],\,x,\,a+b\,\mathrm{ArcCosh}[c\,x]\right]\,\partial_x\,(a+b\,\mathrm{ArcCosh}[c\,x])$

Basis: If $m \in \mathbb{Z}$, then

$$\frac{x^{m-1}\left(m-(m+1)\ c^2\ x^2\right)}{\sqrt{1+c\ x}\ \sqrt{-1+c\ x}} = \frac{1}{b\ c^m}\ \text{Subst}\left[\text{Cosh}\left[-\frac{a}{b}+\frac{x}{b}\right]^{m-1}\left(m-(m+1)\ \text{Cosh}\left[-\frac{a}{b}+\frac{x}{b}\right]^2\right),\ x,\ a+b\ \text{ArcCosh}[c\ x]\right]\ \partial_x\ (a+b\ \text{ArcCosh}[c\ x])$$

Note: Although not essential, by switching to the hyperbolic trig world this rule saves numerous steps and results in more compact antiderivatives.

Rule: If $m \in \mathbb{Z}^+ \wedge -2 \leq n < -1$, then

$$\int x^m\ (a+b\ \text{ArcCosh}[c\ x])^n\, dx$$

$$\rightarrow \frac{x^m\ \sqrt{1+c\ x}\ \sqrt{-1+c\ x}\ (a+b\ \text{ArcCosh}[c\ x])^{n+1}}{b\ c\ (n+1)} + \frac{1}{b\ c\ (n+1)} \int \frac{x^{m-1}\left(m-(m+1)\ c^2\ x^2\right)\ (a+b\ \text{ArcCosh}[c\ x])^{n+1}}{\sqrt{-1+c\ x}\ \sqrt{1+c\ x}}\, dx$$

$$\rightarrow \frac{x^m\ \sqrt{1+c\ x}\ \sqrt{-1+c\ x}\ (a+b\ \text{ArcCosh}[c\ x])^{n+1}}{b\ c\ (n+1)} + \frac{1}{b^2\ c^{m+1}\ (n+1)}\ \text{Subst}\left[\int x^{n+1}\ \text{Cosh}\left[-\frac{a}{b}+\frac{x}{b}\right]^{m-1}\left(m-(m+1)\ \text{Cosh}\left[-\frac{a}{b}+\frac{x}{b}\right]^2\right) dx,\ x,\ a+b\ \text{ArcCosh}[c\ x]\right]$$

Program code:

```
Int[x_^m_.*(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  x^m*Sqrt[1+c*x]*Sqrt[-1+c*x]*(a+b*ArcCosh[c*x])^(n+1)/(b*c*(n+1)) +
  1/(b^2*c^(m+1)*(n+1))*
    Subst[Int[ExpandTrigReduce[x^(n+1),Cosh[-a/b+x/b]^(m-1)*(m-(m+1)*Cosh[-a/b+x/b]^2),x],x],x,a+b*ArcCosh[c*x]] /;
FreeQ[{a,b,c},x] && IGtQ[m,0] && GeQ[n,-2] && LtQ[n,-1]
```

2: $\int x^m\ (a+b\ \text{ArcCosh}[c\ x])^n\, dx$ when $m \in \mathbb{Z}^+ \wedge n < -2$

Derivation: Integration by parts

Basis: $\frac{(a+b\ \text{ArcCosh}[c\ x])^n}{\sqrt{1+c\ x}\ \sqrt{-1+c\ x}} = \partial_x\ \frac{(a+b\ \text{ArcCosh}[c\ x])^{n+1}}{b\ c\ (n+1)}$

Basis: $\partial_x\left(x^m\sqrt{1+c\,x}\,\sqrt{-1+c\,x}\right) == -\frac{m\,x^{m-1}}{\sqrt{1+c\,x}\,\sqrt{-1+c\,x}}+\frac{c^2\,(m+1)\,x^{m+1}}{\sqrt{1+c\,x}\,\sqrt{-1+c\,x}}$

Rule: If $m\in\mathbb{Z}^+\land n<-2$, then

$$\int x^m\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx\,\longrightarrow$$

$$\frac{x^m\sqrt{1+c\,x}\,\sqrt{-1+c\,x}\,(a+b\,\text{ArcCosh}[c\,x])^{n+1}}{b\,c\,(n+1)}+$$

$$\frac{m}{b\,c\,(n+1)}\int\frac{x^{m-1}\,(a+b\,\text{ArcCosh}[c\,x])^{n+1}}{\sqrt{1+c\,x}\,\sqrt{-1+c\,x}}\,dx-\frac{c\,(m+1)}{b\,(n+1)}\int\frac{x^{m+1}\,(a+b\,\text{ArcCosh}[c\,x])^{n+1}}{\sqrt{1+c\,x}\,\sqrt{-1+c\,x}}\,dx$$

Program code:

```
Int[x_^m_.*(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  x^m*Sqrt[1+c*x]*Sqrt[-1+c*x]*(a+b*ArcCosh[c*x])^(n+1)/(b*c*(n+1)) +
  m/(b*c*(n+1))*Int[x^(m-1)*(a+b*ArcCosh[c*x])^(n+1)/(Sqrt[1+c*x]*Sqrt[-1+c*x]),x] -
  c*(m+1)/(b*(n+1))*Int[x^(m+1)*(a+b*ArcCosh[c*x])^(n+1)/(Sqrt[1+c*x]*Sqrt[-1+c*x]),x] /;
FreeQ[{a,b,c},x] && IGtQ[m,0] && LtQ[n,-2]
```

3: $\int x^m (a+b\,\mathrm{ArcCosh}[c\,x])^n\,dx$ when $m\in\mathbb{Z}^+$

Derivation: Integration by substitution

Basis: $F[x] == \frac{1}{b\,c}\,\mathrm{Subst}\left[F\left[\frac{\mathrm{Cosh}\left[-\frac{a}{b}+\frac{x}{b}\right]}{c}\right]\mathrm{Sinh}\left[-\frac{a}{b}+\frac{x}{b}\right],\,x,\,a+b\,\mathrm{ArcCosh}[c\,x]\right]\,\partial_x\,(a+b\,\mathrm{ArcCosh}[c\,x])$

Note: If $m\in\mathbb{Z}^+$, then $x^n\,\mathrm{Cosh}\left[-\frac{a}{b}+\frac{x}{b}\right]^m\mathrm{Sinh}\left[-\frac{a}{b}+\frac{x}{b}\right]$ is integrable in closed-form.

Rule: If $m\in\mathbb{Z}^+$, then

$$\int x^m (a+b\,\mathrm{ArcCosh}[c\,x])^n\,dx \longrightarrow \frac{1}{b\,c^{m+1}}\,\mathrm{Subst}\left[\int x^n\,\mathrm{Cosh}\left[-\frac{a}{b}+\frac{x}{b}\right]^m\mathrm{Sinh}\left[-\frac{a}{b}+\frac{x}{b}\right]dx,\,x,\,a+b\,\mathrm{ArcCosh}[c\,x]\right]$$

Program code:

```
Int[x_^m_.*(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  1/(b*c^(m+1))*Subst[Int[x^n*Cosh[-a/b+x/b]^m*Sinh[-a/b+x/b],x],x,a+b*ArcCosh[c*x]] /;
FreeQ[{a,b,c,n},x] && IGtQ[m,0]
```

U: $\int (d\,x)^m (a+b\,\mathrm{ArcCosh}[c\,x])^n\,dx$

Rule:

$$\int (d\,x)^m (a+b\,\mathrm{ArcCosh}[c\,x])^n\,dx \longrightarrow \int (d\,x)^m (a+b\,\mathrm{ArcCosh}[c\,x])^n\,dx$$

Program code:

```
Int[(d_.*x_)^m_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  Unintegrable[(d*x)^m*(a+b*ArcCosh[c*x])^n,x] /;
FreeQ[{a,b,c,d,m,n},x]
```